
2 INTRODUCTION 

Microsoft has a long history with offering optimization tools. Inside Excel is Solver which can solve linear 

programming problems, mixed-integer programming problems and non-linear problems (Fylstra, Lasdon, Watson, 

& Warren, 1998). Arguably this makes Microsoft the most successful vendor of optimization software with over 

500 million copies distributed
1
. 

The Excel solver is using cells and cell-references to formulate and implement optimization models. This has an 

obvious advantage: Excel users are directly familiar with this structure and can build optimization models without a 

steep learning curve. The direct implementation of a model in Excel has of course numerous benefits such as 

availability of data manipulation tools and report writing facilities including the availability of numerous built-in 

functions, dynamic graphs, and pivot tables. There are also some serious disadvantages:  spreadsheet modeling is 

error prone
2
, the model lacks structure and dimensionality and we are missing out on a short, compact 

representation of the model that allows us to think and talk about it and to share it with colleagues. 

Modeling is difficult. Practical optimization models are often messy with ad-hoc and ǳƴǎǘǊǳŎǘǳǊŜŘ άōǳǎƛƴŜǎǎ ǊǳƭŜǎέ 

(to use a buzzword). This means that any help in adding structure to the model is very welcome. Opposed to usual 

computer programming where we can often break down a complicated part into smaller more manageable pieces 

(Wirth, 1971) we deal essentially with systems of simultaneous equations. In this situation stepwise refinement 

into smaller entities is often not a viable strategy: we need to look at the model as a whole. A high-level modeling 

language can help here: it can provide a compact representation of the model that allows the modeler to view and 

comprehend a complete model. This in turn will allow the modeler to adapt and maintain the model with much 

more ease than otherwise possible. Compared to using a modeling language, the use of a solver API (Application 

Programming Interface) is really a step backward: it will create a more cluttered, wieldy expression of a model that 

makes maintenance and experimentation more difficult, time-consuming and expensive. For very structured or 

very small models this may be not prohibitive, but for large, complex models a good modeling language is an 

invaluable tool that can make a modeler more efficient by orders of magnitude. 

The new Microsoft Solver Foundation product is what we focus on in this document. MSF consists of a number of 

modules: 

¶ Solvers (LP, MIP, CSP) 

¶ OML: an equation based modeling language 

¶ !tLΩǎΥ ǇǊƻƎǊŀƳƳƛƴƎ ƛƴǘŜǊŦŀŎŜǎ ŀƭƭƻǿƛƴƎ ǇǊƻƎǊŀƳƳŜǊǎ ǘƻ ǘŀƭƪ ǘƻ {ƻƭǾŜǊ Foundation services 

¶ Solver plug-in capabilities: external solvers can be hooked up. With version 1.1 the state-of-the-art Gurobi 

MIP solver has become the default MIP solver. It is accessed through this mechanism. 

¶ An Excel based framework to develop and solve OML models 



 

We will concentrate on the modeling language OML and the Excel application framework. 

2.1 MODELING LANGUAGE VS. API 

A model can be built using a modeling language or using a traditional programming language such as C. In the 

latter case one can use a solver API (Application Programming Interface) to assemble the model.  We see that 

many beginners in modeling are attracted to using the API, especially if they have a background and experience in 

computer programming. In my opinion this API-appetite is often unwarranted: if the model is not either very small 

or large but very structured, expressing the model in a specialized modeling language is by far preferable in terms 

of productivity, performance, quality and maintainability of the model.   

Developing a model in a modeling language is often much more efficient.  First of all, a model expressed in a 

modeling language is much more compact. The same model in a programming language will require many more 

lines of code.  Further we often see modelers struggling with low level issues like memory allocation, pointer 

problems and obscure linker messages that simply do not occur when using a modeling language. The gain in 

productivity can be used spend more time to improve the model formulation. 

Large, difficult models require many revisions and experiments to achieve best performance (speed and reliability). 

Different formulations can lead to large differences in performance, so it is beneficial if it is easy to try out 

different formulations quickly.  Here a modeling language shines compared to a traditional programming language. 

The most well-known modeling languages are GAMS and AMPL. They are both fairly complex systems, and there is 

a learning curve before you are comfortable with these languages. But once you mastered them, you can build 

large, complex, maintainable models in a controlled fashion. OML is a much simpler language. Much of the 

complexity (such as any data manipulation) is moved away from the modeling language to the environment where 

OML is called from. This can be a C# program, or in the case of this paper Excel. This approach comes with some 

advantages (a simpler, cleaner modeling language) and disadvantages (more complex and precise data preparation 

is needed before we can pass data on to the OML model to form a complete model instance). In this paper we will 

explore some of these issues.  

We will focus on OML as used in the Excel plug-in. The tight integration between Excel and OML gives a rich but 

unstructured environment for data handling and reporting, a small, limited modeling language, a build-in scripting 

language (VBA) and enough widgets such as buttons to create mini-applications.  

2.2 A TRANSPORTATION MODEL 

The transportation model is among the simplest Linear Programming models we can present. 

We want to minimize shipping cost while obeying demand and supply restrictions. The mathematical model can be 

stated as: 

minВ ὧὭ,Ὦ ὼὭ,ὮὭ,Ὦ

ВὼὭ,Ὦ ὨὮ  ᶅ ὮὭ

ВὼὭ,Ὦ ίὭὮ  Ὥᶅ

ὼὭ,Ὦ 0

  

 



 

Here x is the decision variable and c, d, and s are parameters. The difference between a parameter and a decision 

variable is an important one. A parameter is a constant during the solution of the model: it will not be changed by 

the solver. A variable will be changed by the solver: hopefully it will return the best possible values for the decision 

variables.  

In this section we will compare the OML representation of this model to two alternatives: a GAMS formulation and 

an implementation in Excel Solver. 

2.2.1 A GAMS REPRESENTATION 

The first model in the Model Library from GAMS is a simple example of this problem, based on the famous text 

book (Dantzig, 1963)
3
. The complete model looks like: 

$Title  A Transportation Problem (TRNSPORT,SEQ=1)  
$Ontext  
 
This problem finds a least cost shipping schedule that meets 
requirements at markets and supplies at factories.  
 
 
Dantzig, G B, Chapter 3.3. In Linear Programming and Extensions.  
Princeton University Press, Princeton, New Jersey, 1963.  
 
This formulation is described in detail in:  
Rosenthal, R E, Chapter 2: A GAMS Tutorial. In GAMS: A User's Guide.  
The Scientific Press, Redwood City, California, 1988.  
 
The line numbers will not match those in the book because of these  
comments. 
 
$Offtext  
 
  Sets  
       i   ' canning plants '    / seattle, san - diego /  
       j   ' markets '           / new - york, chicago, topeka / ;  
 
  Parameters  
 
       a(i)  ' capacity of plant i in cases '  
         /    seattle     350  
              san- diego   600  /  
 
       b(j)  ' demand at market j in cases '  
         /    new - york    325  
              chicago     300  
              topeka      275  / ;  
 
  Table d(i,j)  ' distance in thousands of miles '  
                    new- york       chicago      topeka  
      seattle          2.5           1.7          1.8  
      san- diego        2.5           1.8          1.4  ;  
 
  Scalar f  ' freight in dollars per case per thousand miles '   /90/ ;  



 

 
  Parameter c(i,j)  ' transport cost in thousands of dollars per case '  ;  
 
            c(i,j) = f * d(i,j) / 1000 ;  
 
  Variables  
       x(i,j)  ' shipment quantiti es in cases '  
       z       ' total transportation costs in thousands of dollars '  ;  
 
  Positive Variable x ;  
 
  Equations  
       cost        ' define objective function '  
       supply(i)   ' observe supply limit at plant i '  
       demand(j)   ' satisfy demand at market j '  ;  
 
  cost ..        z  =e=  sum((i,j), c(i,j)*x(i,j)) ;  
 
  supply(i) ..   sum(j, x(i,j))  =l=  a(i) ;  
 
  demand(j) ..   sum(i, x(i,j))  =g=  b(j) ;  
 
  Model transport /all/ ;  
 
  Solve transport using lp minimizing z ;  
 
  Display x.l, x.m ;  
 

The sets indicate collections of strings that we use for indexing. GAMS uses string as vehicle for indexing vectors 

ŀƴŘ ƳŀǘǊƛŎŜǎΦ ¢Ƙƛǎ Ƙŀǎ ǎƻƳŜ ŀŘǾŀƴǘŀƎŜǎΥ ƛǘ ƳŀƪŜǎ ǘƘŜ ƳƻŘŜƭ ŜŀǎƛŜǊ ǘƻ ǊŜŀŘ όǇƭŀƴǘ Ψ{ŜŀǘǘƭŜΩ ƛǎ ƳƻǊŜ ŘŜǎŎǊƛǇǘƛǾŜ ǘƘŀƴ 

plant 1), and it makes it unattractive to use index arithmetic in cases where this may not be needed. The latter is 

also a negative as the obvious disadvantage is that it makes index arithmetic more complicated where we can 

legitimately use it. 

Parameter, scalar and table statements are used to specify parameters. Parameters can be changed inside the 

GAMS model (using assignment statements) but not by the solver. During the SOLVE statements parameters are 

constants. GAMS allows for convenient data entry: only the nonzero elements need to be provided in parameter 

and table statements.  Data manipulation is done by assignment statements, like c(i,j) = f * d(i,j) / 

1000 which can be interpreted as an implicit loop.   

The optimization model itself starts with variable and equation declarations.  By default variables are free, i.e. they 

are allowed to assume positive and negative values. With Positive Variable x  we impose a lower bound of 

zero. The equations are declared with a somewhat peculiar syntax. Equality is denoted by =e= while =l=  and =g= 

are less-than-or-equal and greater-than-or-equal constraints. Note that each constraint is actually a block of 

constraints. E.g. constraint demand(j)  implements three constraints  because set j has three elements.  

It is important to understand the difference between assignment statements and equations. Assignments are 

executed by GAMS itself in order as they appear, while equations are passed on the solver and must hold 

simultaneously. In programming language parlor we say that data manipulation (i.e. assignments) is procedural 

while model equations are declarative. 



 

Finally we have a model and solve statement and the results are displayed. GAMS has the notion of an objective 

variable opposed to an objective function. In practice this is not a problem: just place your objective in an equality 

constraint, and optimize the corresponding variable. Note that x.l, x.m  indicates we want to see the optimal 

level values and the optimal marginal values (or reduced cost) of x.  

The results of a GAMS job are written to a listing file.  The listing file will contain: 

¶ A source listing of the model. This can be useful to find the location of syntax errors or run-time errors. 

¶ A listing of individual rows and columns generated by the model, i.e. the expanded model. This is useful 

for debugging.  

¶ A section with messages from the solver. Hopefully it will say OPTIMAL. 

¶ The solution: rows and columns. Both level values and marginals are printed. Marginals are reduced costs 

for variables and duals for equations. 

¶ The output of display statements. 

GAMS has built-in facilities for report-writing: we can use data-manipulation on solution vectors, and display the 

final results. 

For large models, GAMS has a number of facilities: 

¶ !ƭƭ Řŀǘŀ ǎǘǊǳŎǘǳǊŜǎ ŀǊŜ ǎǇŀǊǎŜΥ ƴƻ ǎǘƻǊŀƎŜ ŦƻǊ ȊŜǊƻΩǎ 

¶ Ϸ ŎƻƴŘƛǘƛƻƴǎ ŀƭƭƻǿ ƛƳǇƭŜƳŜƴǘƛƴƎ ΨǎǳŎƘ ǘƘŀǘΩ ƻǇŜǊŀǘƛƻƴǎ ƻƴ ǎŜǘǎ 

¶ Abort statement for error checking 

¶ Loop statement to handle multiple solve statements, e.g. to implement heuristics 

¶ GAMS comes with an IDE (under Windows) 

The integration with Excel is limited. There is an external program (gdxxrw.exe) that allows for exchanging data 

between GAMS and Excel, but to use this from an active Excel spreadsheet is difficult. It requires a fairly large 

amount of VBA code to run GAMS from Excel. See http://www.amsterdamoptimization.com/packaging.html. 

2.2.2 AN EXCEL SOLVER APPROACH 

The traditional way to model this problem in Excel is to use Solver. First we setup the data. This includes unit cost 

coefficients c, supply capacity s and demand data d. In our case we calculate the cost coefficients from unit 

transportation cost and a distance table. 

http://www.amsterdamoptimization.com/packaging.html


 

 

Figure 1. Data for the TRNSPORT model 

The cost data are calculated by the formula: 

ὧὭ,Ὦ=
ὪẗὨὭ,Ὦ

1000
 

E.g. cell E10 has formula =E6*Freight/1000 . Now we setup a table where the shipments will be placed. 

 

Figure 2. Derived data 

The cells E16:G17 correspond to ὼὭ,Ὦ. In addition we calculate the row and column sums. E.g. cell H16 has formula 

=SUM(E16:G16). The total cost are calculated by: =SUMPRODUCT(E10:G11,E16:G17) which represents the 

expression 

ὧὭ,ὮὼὭ,Ὦ
Ὥ,Ὦ

 

A complete view of the formulas is: 



 

 

Figure 3. Excel Solver model formulation 

The model can now be specified in the solver  

 

Figure 4. Setup Solver 

Here we specify: 

¶ The objective is total cost at cell J16 

¶ The variables ὼὭ,Ὦ are located in the table E16:G17 

¶ The constraints are formed by calculating the sums ВὼὭ,ὮὮ  and ВὼὭ,ὮὭ  and comparing those quantities 

with the available supply and required demand. 

¶ The conditions ὼὭ,Ὦ 0 ŀǊŜ ǎǇŜŎƛŦƛŜŘ ƛƴ ǘƘŜ ƻǇǘƛƻƴǎ ŘƛŀƭƻƎ ǿƘŜǊŜ ǿŜ ŎƘŜŎƪ ǘƘŜ άŀǎǎǳƳŜ ƴƻƴ-ƴŜƎŀǘƛǾŜέ 

option. 

¶ Lƴ ŀŘŘƛǘƛƻƴ ǿŜ ǎǇŜŎƛŦȅ ǘƘŜ ƳƻŘŜƭ ǘƻ ōŜ ƭƛƴŜŀǊ ōȅ ŎƘŜŎƪƛƴƎ άŀǎǎǳƳŜ ƭƛƴŜŀǊ ƳƻŘŜƭέΦ 

The disadvantage of this approach is visible here: it is difficult to recognize the model 

minВ ὧὭ,Ὦ ὼὭ,ὮὭ,Ὦ

ВὼὭ,Ὦ ὨὮ  ᶅ ὮὭ

ВὼὭ,Ὦ ίὭὮ  Ὥᶅ

ὼὭ,Ὦ 0

  

 



 

in this spreadsheet. For a small or well-structured model this may not be a problem, but for more complex models, 

the lack of structure will be a major obstacle to build and maintain models efficiently. Indeed I have converted a 

number of Excel Solver models, and the majority of them contained errors such that I could not reproduce the 

results when using a GAMS model (in different words: the Excel solution was wrong). 

 

 

Figure 5. Excel Solver Options 

The formulation can be improved by using named ranges. 

The optimal values are updated in the spreadsheet. In addition the solver can generate a solution report as 

follows: 



 

 

Figure 6. Excel Solver Report 

2.2.3 THE OML IMPLEMENTATION 

The OML model to represent this mathematical model could look like: 

Model[  
 
  Parameters[Sets,Plants,Markets],  
  Parameters[Reals,Capacity[Plants],Demand[Markets],Cost[Plants,Markets]],  
 
  Decisions[Reals[0,Infinity],x[Plants,Markets],TotalCost],  
 
  Constraints[  
      TotalCost == Sum[{i,Plants},{j,Markets},Cost[i,j]*x[i,j]],  
      Foreach[{i,Plants}, Sum[{j,Markets},x[i,j]]<=Capacity[i]],  
      Foreach[{j,Markets}, Sum[{i,Plants},x[i,j] ]>=Demand[j]]  
  ],  
 
  Goals[Minimize[TotalCost]]  
]  
 

Figure 7. TRNSPORT model in OML 

This model only has the declarative part of the corresponding GAMS model:  declaration statements and equation 

definitions.  



 

The sets are declared as part of parameter statement: they are parameter of a special type. The variables are 

called Decisions, and we can specify bounds on them in the declaration. By default OML assumes variables are 

free, just like GAMS (and unlike many other mathematical programming languages). Finally we specify the 

equations. 

The data and the data manipulation need to be handled somewhere else. In this case we choose the spreadsheet 

interface. The screen-shot below shows how the data can be organized and entered in Excel: 

 

Figure 8. Data Binding for the TRNSPORT model 

The data is organized in a table format in this example. The data binding is slightly different for table formatted 

data or block formatted data. 

The results are written to a separate sheet, formatted by Solver Foundation. 



 

 

Figure 9. Solver Foundation Solution Report 

This modeling approach is a little bit between GAMS and the old Solver based structure: the model is specified in 

an equation-based modeling language OML, while all data manipulation and reporting is done in Excel. 

2.2.4 MICROSOFT SOLVER FOUNDATION 

The Solver Foundation comes standard with a number of solvers: 

¶ LP solvers: there are several linear programming solvers: an interior point solver and a simplex based 

primal and dual solvers 

¶ MIP solver: the Mixed Integer solver allows discrete variables. From version 1.1 the default MIP solver is 

Gurobi. 

¶ QP solver: the interior point solver can also solve QP problems, i.e. models with a quadratic objective 

¶ CSP solver: a solver for constraint programming problems is available 

Compared to GAMS and AMPL we miss a general non-linear programming functionality ς MSF contains an 

unconstrained NLP solver but this is not supported by OML. On the flip-side MSF offers a CSP solver. GAMS has no 

facilities to support CSP models, and in the case of AMPL some work has been done in this direction (Fourer, 1998), 

but to my knowledge this is not available in the official distribution. An example of a general purpose modeling 

language with support for constraint programming is OPL (van Hentenryck, 1999). 

Now we have seen how the trnsport  model can be implemented in three different ways, we will continue with 

the OML/Excel combination. We will explain the language briefly, and then provide a number of annotated 

examples. 

3 OML THE LANGUAGE 

OML (Optimization Modeling Language) is the modeling language of the Microsoft Solver Foundation. 

The basic structure of an OML model is: 

-ÏÄÅÌǁ0ÁÒÁÍÅÔÅÒÓǁƛǂƗ$ÅÃÉÓÉÏÎÓǁƛǂƗ#ÏÎÓÔÒÁÉÎÔÓǁƛǂƗ'ÏÁÌÓǁƛǂǂ 

Note that OML is case-ǎŜƴǎƛǘƛǾŜ ǘƘǊƻǳƎƘƻǳǘΣ ƛΦŜΦ άaƻŘŜƭέ ƛǎ ƴƻǘ ǘƘŜ ǎŀƳŜ ŀǎ άƳƻŘŜƭέΦ 



 

   

3.1 PARAMETERS 

The Parameters section declares sets and parameters. We will discuss here only the declarations. The 

corresponding data typically is read from a spreadsheet, which we will discuss in section Data Binding. 

3.1.1 SETS 

Sets are declared in OML but they cannot be populated in OML: all content comes from data binding. 

A set is declared as: 

Parameters[Sets,Plants,Markets]  

We now have can use two sets for indexing. The actual elements for these sets will need to come from Excel 

through data binding (this is explained later). 

We can split a declaration in pieces. I.e. the statements 

Parameters[Sets,I],Parameters[Sets,J]  

and 

Parameters[Sets,I,J]  

are identical. 

In some cases we want to be able to perform arithmetic on set elements. We can force set elements to be integers 

using the syntax: 

 Parameters[Sets[Integers],I,J]  

3.1.2 SCALARS 

A scalar can be declared as follows: 

Parameters[Reals,a=10]  

It is possible to add bounds to the domain. E.g. 

Parameters[Reals[0,100],a=110]    // illegal: domain violation  

will give an error. The same is true for 

Parameters[Integer s,a= 0.5 ]    // illegal: domain violation  

The following statements are identical: 

Parameters[Real s,a= 0.5 ]    // fraction  



 

and 

Parameters[Real s,a= 1/2 ]    // alternative notation for fraction  

3.1.2.1 RESTRICTIONS 

Scalars cannot be read from a spreadsheet using the syntax above. If you need to read a scalar from a spreadsheet 

you need to declare 

Parameters[Integer , N[] ]    // read scalar from spreadsheet  

Unfortunately, scalars cannot be assigned a constant expression: 

 Parameters[Integers,N=10],  

 Parameters[Integers,N1=N+1]  // Illegal, no expressions allowed  

 Parameters[Integers[0,N],M ]  // Illegal, bounds need to be literal numbers  

You will need to use (N+1) in the model or import N1[] from the spreadsheet. 

3.1.3 INDEXED PARAMETERS 

Indexed parameters (e.g. vectors and matrices) need to be imported from the spreadsheet. The values cannot be 

specified directly in OML. For more information see the section on Data Binding. 

Parameters[Sets,I,J],  
Parameters[Reals,v[I],A[I,J]],  

 

The domain can be tightened if needed: 

Parameters[Integers[1,5],w[I]]  

This will check if the data for w[i] is integer values and between 1 and 5. If the data violates this, an error will be 

issued and the model will not be solved.  

Note: we can use ƵInfinity  and Infinity  in the bound specification.  

3.1.3.1 RESTRICTIONS 

It is not possible to use a parameter as bound on a domain: 

Parameters[Integers,N=10],  
Parameters[Reals[0,N],v[I],A[I,J]],  // Illegal  use of N  
 

3.2 DECISIONS 

The Decisions section declares the variables. It also specifies bounds on the variables and their type (continuous, 

discrete). 



 

Here are some examples: 

Decisions[Reals,x],  // a free scalar variable  

 Decisions[Reals[0,Infinity],y[I,J]], // a non - negative set indexed variable  

 Decisions[Integers,Foreach[{i,N},d[i]]],  

                                  ƳƳ ÆÒÅÅ ÉÎÔÅÇÅÒ ÖÁÒÉÁÂÌÅÓ ÄǁʣǂƗƛƗÄǁ.- 1]  

If ȅƻǳ ŘƻƴΩǘ ǎǇŜŎƛŦȅ ōƻǳƴŘǎ ha[ ǿƛƭƭ ŀǎǎǳƳŜ ȅƻǳ ŘŜŀƭ ǿƛǘƘ ŦǊŜŜ ǾŀǊƛŀōƭŜǎΦ This is like GAMS but unlike many other 

modeling systems that have non-negative variables as the default.  

3.2.1 DOMAINS 

There are the following types (domains): 

¶ Reals: for continuous variables 

¶ Integers: for integer and binary variables. If you need to use binary variables or zero-one variables use a 

lower bound of zero and an upper bound of one: Decisions[Integers[0,1],x[I,J]]
4
. 

¶ Boolean. These are variables indicating true or false. Note, that these are not the same as binary 

variables. In many cases one would prefer to use a binary variable as they allow numerical expressions in 

equations. 

3.2.2 FOREACH 

The Foreach construct has a number of forms: 

¶ &ÏÒÅÁÃÈǁǅÉƗ.ǆƗƛǂ  loops i=0,1,..N-1. Note that the loop is zero-based. 

¶ &ÏÒÅÁÃÈǁǅÊƗËƗ.ǆƗƛǂ which loops j=k,k+1,..,N-1. Note that the loop does not include N. 

¶ &ÏÒÅÁÃÈǁǅÉƗ)ǆƗƛǂ  loops over set I. 

Note that &ÏÒÅÁÃÈǁǅÉƗ.ǆƗƛǂ is identical to &ÏÒÅÁÃÈǁǅÉƗʣƗ.ǆƗƛǂ. 

The Foreach construct can be used to declare indexed variables one by one: 

Parameters[Integers,N=10],  

Decisions[Reals, Foreach[{i,N},x[i]]]  

3.2.3 FILTEREDFOREACH 

The FilteredForeach construct is an extension of the Foreach expression: it adds a condition over which the loop is 

executed. I.e. FilteredForeach[{ƛΣbϒΣƛΗҐоΣΧϐ ƭƻƻǇǎ ƛҐлΣмΣнΣпΣрΣΧΣb-1.  

In some cases you may want to generate not a full x[i,j] but only a subset, e.g. only x[i,j] for i>j. This can be 

specified as: 

Parameters[Integers,N= 3],  

Decisions[Reals, Foreach[{i,N},{j,i},x[i,j]] ],  

    // strictly  lower - triangular matrix  

If N=3, this will generate the variables x[1,0],x[2,0],x[2,1]. 



 

LŦ ȅƻǳ ǿƻǳƭŘ ƭƛƪŜ ǘƻ ƎŜƴŜǊŀǘŜ ȄώƛΣƧϐΣ ƛҐмΣΦΦΦΣbΣ ƧҖƛΣ (lower triangular including diagonal, one-based indexing) then we 

have 

 Decisions[Reals, Foreach[{i,1,N+1},{j,1,i+1},x[i,j]]  

An alternative formulation would be: 

 Decisions[Reals,Foreach[{i,1,N+1},FilteredForeach[{j,1,N+1},j< =i,x[i,j]]]],  

 // does not work in 1.1 (worked in 1.0)  

Note that you cannot write: 

Decisions[Reals, Filtered Foreach[{i,1, N+1},{j,1,N+1},j< =i,x[i,j]]],  // illegal  

In general OML models look better and are simpler if zero-based indexing is used. 

The Filtered Foreach []  construct does not allow a condition to depend on a variable. If a constraint is to be 

activated depending on a variable, the Implies []  construct can be used. 

3.3 CONSTRAINTS 

Constraints are equations (equalities or inequalities) that put extra conditions on the solution. I often will use the 

terms constraint, equation and row interchangeably, all indicating the same concept. 

The transportation model from section 1 has the following constraint section: 

Constraints[  
      TotalCost == Sum[{i,Plants},{j,Markets},Cost[i,j] *x[i,j]],  
      Foreach[{i,Plants}, Sum[{j,Markets},x[i,j]]<=Capacity[i]],  
      Foreach[{j,Markets}, Sum[{i,Plants},x[i,j]]>=Demand[j]]  
  ],  
 

Figure 10. Constraints from the TRNSPORT model 

This example directly corresponds to the underlying mathematical model: 

For a linear programming or mixed-integer programming model, all these constraints have to be linear. 

3.3.1 SUMMATIONS 

The Sum expression is an important tool in modeling optimization models. Summations over set indexed 

expressions are very straightforward: 3ÕÍǁǅÉƗ)ǆƗƛǂ. A double summation in this context is also obvious: 

Sum[{i,I},{ j , JǆƗƛǂ. 



 

The summation over integers is like the Foreach. 3ÕÍǁǅÉƗ.ǆƗƛǂ ƛǎ ǎǳƳƳƛƴƎ ƻǾŜǊ ƛҐлΣмΣΧΣb-1. The version with a 

lower limit 3ÕÍǁǅÊƗËƗ.ǆƗƛǂ ǿƛƭƭ ǎǳƳ ƻǾŜǊ ƧҐƪΣƪҌмΣΧΣb-1. This means that Sum[{k,1,3},1] = 2 .  

 In addition there is a FilteredSum []  variant, which adds a condition to the summation. E.g. 

FilteredSum[{k,1,4},k!=2,k] = 4 . The FilteredSum []  construct does not allow that a condition 

depends on a variable, even in case of a CSP model.  Typically in that case one would use an ordinary Sum with a 

factor AsInt[condition] . 

3.3.2 ALL-DIFFERENT CONSTRAINTS 

In some constraint programming problems, the following condition may appear: 

X[i] = k, where k is different for each i 

This is difficult to model efficiently in a MIP (Williams & Yan, 2001), but can be easily expressed using the Unequal  

construct. The above condition can be stated as: 

 Unequal[Foreach [{i,N},x[i]]]  

3.3.3 IMPLIES 

¢ƘŜ ŎƻƴǎǘǊǳŎǘ άƛŦ όŎƻƴŘƛǘƛƻƴύ ǘƘŜƴ ŎƻƴǎǘǊŀƛƴǘέ Ŏŀƴ ōŜ ŦƻǊƳǳƭŀǘŜŘ ǳǎƛƴƎ Implies : 

 Implies[x==1,y<=10]  

This is only available for CSP models. Example: 

Foreach[{i,Tiles},Implies[b[i],x[i] <=N- Side[i] & y[i]<=N - Side[i]]]  

3.3.4 AND, OR 

In CSP models we can use !ÎÄǁÅØÐʦƗÅØÐʧƗƛǂ and /ÒǁÅØÐʦƗÅØÐʧƗƛǂ. They can also be written using infix 

operators: exp1 & exp2  and exp1 | exp2 . The example in the previous paragraph shows an application. 

Another example is: 

Foreach[{i,T},{j,i+1,T},Implies[b[i]&b[j],  
             x[i]>=x[j]+Side[j]  |  
             x[i]+Side[i]<=x[j]  |  
             y[i]>=y[j]+Side[j]  |  
             y[i]+Side[i]<=y[j]  
         ]]  

 

3.4 DATA BINDING 

The OML language requires that all data is specified outside the model (except for an occasional scalar constant). 

The concept to move data from the spreadsheet to the model instance is called Data Binding. 



 

3.4.1 SCALARS 

To import a scalar we just need to know the address of the cell where it is located: 

 

Figure 11. A scalar parameter 

In this case this is Sheet1!C3.  

In the OML model we specify: 

Parameters[Reals,s[]],  

where it is important to use [] in the declaration. [] indicates: the symbol has zero dimensions. Then in the Data 

Binding dialog we can use: 

 

Figure 12. Data binding for a scalar 

Note: we did not check άaȅ Řŀǘŀ ƛǎ ŀ ǘŀōƭŜέΦ To test this we can use the minimal model: 



 

 

Figure 13. A test model 

Note that we have to use a constraint. My first attempt was to write this as a fixed variable: 

Model[  
 
   Parameters[Reals,s[]],  
 
   Decisions[Reals[s[],s[]],x]  // not allowed  
 
]  

This construct is not allowed: we cannot fix the variable by setting its lower and upper bound to s[].

We can dynamically size a variable based on a bound parameter: 

Model[  
  
  Parameters[Integers,n[]],   // retrieve through data binding  
  Decisions[Reals,Foreach[{i,n[]},x[i]]]   // OK since 1.1  
  
]  
 

3.4.2 VECTORS 

A vector can be imported as follows. The range will become not a single cell but a range indicating the complete 

vector: 



 

 

In this case that will be Sheet1!F4:F7. A minimal model to read this is: 

 

The set I will automatically be populated with the index positions 0,1,2,3. This can be seen in the solution report: 

Solver Foundation Results 

Name Value 

  Solution Type Optimal 

  x[0] 1 

  x[1] 2 

  x[2] 3 

  x[3] 4 

  

3.4.3 EMPTY CELLS 



 

In Excel we often can work with empty cells. E.g. the =SUM formula will not be bothered by empty cells, they will 

just be skipped. In OML, empty cells cause an error. 

E.g. in Excel we can do: 

 

Excel is fine with the empty cell at C7 but if we try to emulate this in OML we see: 

 

As we will see in the examples this is can cause some headaches when dealing with larger models. 



 

3.4.4 MATRICES 

This approach extends to matrices. For the data: 

 

we can use the range Sheet1!F5:H6. The model needs to be changed to: 

Model[  
    
    Parameters[Sets,I,J],  
    Parameters[Reals,v[I,J] ],  
   
    Decisions[Reals,x[I,J]],  
    Constraints[Foreach[{i,I},{j,J},x[i,j]==v[i,j]]]  
 
]  
 

 Lƴ ǘƘŜ Řŀǘŀ ōƛƴŘƛƴƎ ŘƛŀƭƻƎ ōƻȄ ŀƎŀƛƴ ƳŀƪŜ ǎǳǊŜ ǘƘŀǘ άaȅ Řŀǘŀ ƛǎ ŀ ǘŀōƭŜέ ƛǎ ƴƻǘ ŎƘŜŎƪŜŘΦ  The solution will now look 

like: 

Solver Foundation Results 

Name Value 

 Solution Type Optimal 

 x[0,0] 0.661085096 

 x[0,1] 0.712827648 

 x[0,2] 0.719861886 

 x[1,0] 0.092979643 

 x[1,1] 0.332064994 

 x[1,2] 0.793925364 

 
 

bƻǘŜ ŀƎŀƛƴ ǘƘŀǘ ŜƳǇǘȅ ŎŜƭƭǎ ŀǊŜ ƴƻǘ ŀƭƭƻǿŜŘ ƛƴ ha[Υ ȅƻǳ ƴŜŜŘ ǘƻ ǎǇŜŎƛŦȅ ȊŜǊƻΩǎ ŜȄǇƭƛŎƛǘƭȅ ƛƴ ŀƭƭ ŜƳǇǘȅ cells. 

3.4.5 DATA TABLES  

5ŀǘŀ ŦƻǊƳŀǘǘŜŘ ŀǎ ŀ ŘŀǘŀōŀǎŜ ǘŀōƭŜ Ŏŀƴ ōŜ ǊŜŀŘ ǳǎƛƴƎ ǘƘŜ άaȅ Řŀǘŀ ƛǎ ŀ ǘŀōƭŜέ ƻǇǘƛƻƴΦ ¢Ƙƛǎ ǿƛƭƭ ŀƭƭƻǿ ȅƻǳ ǘƻ ǎǇŜŎƛŦȅ 

keys (they will becomes index sets in OML) and values. For an example we go back to our transportation model 

from section 2.2.3. We see there: 



 

 

!ƭƭ Řŀǘŀ ƛǎ ƻǊƎŀƴƛȊŜŘ ƛƴ ǘŀōƭŜǎ ǿƘŜǊŜ ŜŀŎƘ ŎƻƭǳƳƴ ŎƻǊǊŜǎǇƻƴŘǎ ǘƻ ŀ άŘŀǘŀōŀǎŜ ŦƛŜƭŘέ ŀƴŘ ŜŀŎƘ Ǌƻǿ ƛǎ ŀ ǊŜŎƻǊŘΦ Lƴ 

this case the Data Binding dialog will look like: 

 

I.e. in the Demand table, the values are in column Demand (that will also become the name of the parameter) and 

the keys (set elements) are in column Markets. 

From a larger table we can select a Value column. E.g. in the table with Distance and Cost data we want to extract 

the parameter Cost[Plants,Markets]. This can be achieved by: 



 

 

Note that in the dialog we specified the whole range B12:E18. The system will ignore unused parts (like the column 

Distance), and it will show the actual used range as: 

 

Indeed column D is not used in the data binding for Cost[Plants,Markets]. 

From this example you can also see that we used the sets Plants and Markets several times. So who determines 

what these sets will contain. The answer is that they will contain the union of all elements used.  

Note that sets imported this way need to be strings. If the cell has a number OML will complain about a conversion 

error. 

3.4.6 SPARSE TABLES 

OML does not support sparse table. That means that records with a zero value need to be explicitly part of the 

table. 

3.4.7 DATA LAYOUT 

It is important to think carefully about the layout of your data. Below is an example: 

Sparse table 
        

          

          



 

We try to read here a sparse graph 
      

          

          This version does not work very well. We cannot This version is better. We now have a way to  

infer easily which arcs are to be used. 
 

form parameter From[Arc], To[Arc] and 
 So parameter Capacity[From,To] would not be Capacity[Arc] 

   useful because of missing combinations. 
 

      From To Capacity 
  

Arc From To Capacity 
 1 2 15 

  
1 1 2 15 

 1 3 10 
  

2 1 3 10 
 1 4 12 

  
3 1 4 12 

 2 5 5 
  

4 2 5 5 
 2 6 5 

  
5 2 6 5 

 2 7 5 
  

6 2 7 5 
 3 5 6 

  
7 3 5 6 

 3 6 6 
  

8 3 6 6 
 3 7 6 

  
9 3 7 6 

 4 5 12 
  

10 4 5 12 
 5 8 10 

  
11 5 8 10 

 6 8 15 
  

12 6 8 15 
 7 8 15 

  
13 7 8 15 

 
The left table describes a sparse network. Only some arcs (i,j) exist and have a capacity. OML cannot handle this 

data. If we would declare Parameters[Reals, Capacity[From,To]]  then OML assumes all combinations 

(From,To) exist. 

In GAMS we could form the arcs from the sparse capacity table as follows: 

set arcs(i,j);  
arcs(i,j)$capacity(i,j) = yes;  

i.e. an arc exists where we have a capacity. 

In OML we cannot build any sets or parameters from other sets and parameters.  One way to organize the data so 

an OML model can work with it is to have a complete matrix Capacity[From,To] for all combinations (From.To) with 

explicitly Capacity[From,To]=0 when the arc does not exist. For larger instances this may become tedious, 

especially when the graph is very sparse. 

Another approach is shown in the right table. Explicitly introduce arc numbers so we can form the parameter 

Capacity[Arc]. The node-arc incidence can then be formed by parameters From[Arc] and To[Arc].  

The exact form of the input data is important not in small part because OML has no data manipulation facilities 

allowing the modeler to repair or massage ill-fitting data. 

3.4.8 RANGE NAMES 



 

Excel data binding does support range names, although the feedback from the GUI does imply otherwise. Excel 

range names can be global for the whole workbook or work-sheet specific. Even for global names we need to 

specify the sheet name. E.g. if a range N is used to indicate a scalar parameter, we need to specify: Sheet1!N.  If we 

try to specify only N we see: 

 

If we specify Sheet1!N it works. However when we open the binding dialog again, we see: 

 

This is misleading however: the actual underlying range is not Sheet1!$F$5 but rather Sheet1!N. We can see this if 

we change the definition of the range N. If we change the range N to say =Sheet1!$E$5 then (only after pressing 

the Solve button) we see the data binding info is changed: 

 



 

I have not been able to use dynamic ranges with data binding. An example of such a dynamic range is 

=OFFSET($A$1,0,0,COUNT($A:$A),1) . These constructs are used to make ranges automatically expand or 

contract depending on the size of a table. This is convenient if you want to allow a user to add or delete rows, such 

that the rest of the spreadsheet is updated accordingly. With fixed ranges, adding a row may not change the actual 

calculations as the new row is outside the range that is being operated upon. Allowing dynamic ranges makes it 

possible to create better scalable spreadsheet applications. Of course it is always possible to have some VBA code 

that manipulates a named range. 

3.4.9 DEBUGGING 

There is no good facility to debug data binding: there is no display statement and no equation listing or expand 

command to look at equations.  However the following works: introduce extra variables and extra constraints so 

the variables are equal to the imported data: 

 Foreach[{i,I},xdbug[i]==testpar[i]] 

Then in the solution report you can inspect the variable. 

3.4.10 PERFORMANCE 

It looks like the MS Solver Foundation Excel Plug-in is somewhat slow for importing larger data sets (we observed 

this in this note). Here are a few timings that confirm this: 

 

Cells MSF Excel Plugin GAMS gdxxrw 

n=100 0.7 1.2 

n=1000 2.7 1.3 

n=10000 23.0 1.4 

 

OML Model Binding: P<=="Sheet1!$B$2:$CW$101"  

Model[  

Parameters[Sets,I,J],  

Parameters[Reals,P[I,J]],  

Decisions[Reals,d],  

Constraints[d==0]  

]  

 

GAMS Model $call gdxxrw i=data.xlsx par=p rng=A1:CW101  

parameter  p(*,*);  

$gdxin data  

$load p  

  

scalar n;  

n=card(p);  

display n;  

 

http://yetanothermathprogrammingconsultant.blogspot.com/2008/12/mona-lisa-assignment-in-gams.html


 

The MSF code contains a large parameter that is read through data binding, together with a minimal (one variable, 

one constraint) model. The GAMS code calls GDXXRW to read the spreadsheet and then the intermediate GDX file 

is read again by GAMS. We display the cardinality of p as a check we read all numbers. 

4 SOME API NOTES 

L ŘƻƴΩǘ ǿŀƴǘ ǘƻ ŘŜƭǾŜ ǾŜǊȅ ŘŜŜǇ ƛƴǘƻ ǘƘŜ !tLΩǎ ƻŦŦŜǊŜŘ ōȅ a{CΦ Lƴ ƎŜƴŜǊŀƭ Ƴȅ ǾƛŜǿ ƛǎ ǘƘŀǘ ƳƻŘŜƭƛƴƎ ƛƴ ŀ ƳƻŘŜƭƛƴƎ 

language is often to be preferred to writing ŀ άƳŀǘǊƛȄ ƎŜƴŜǊŀǘƻǊέ ƛƴ ŀ traditional programming language. However 

it is useful to discuss some practical cases where we blur the difference a bit: OML can be used from within C# to 

handle cases where the Excel plug-in has limitations. 

4.1 RUNNING OML FROM C# 

A simple way to run a scalar OML model embedded in a C# file is as follows. 

We consider again the simple transportation model from section 2.2. In OML the model looks like: 

Model[  

  Parameters[Sets,Plants,Markets],  

  Parameters[Reals,Capacity[Plants],Demand[Markets],Cost[Plants,Markets]],  

 

  Decisions[Reals[0,Infinity],x[Plants,Markets],TotalCost],  

 

  Constraints[  

     TotalCost == Sum[{i,Plants},{j,Markets},Cost[i,j]*x[i,j]],  

     Foreach[{i,Plants}, Sum[{j,Markets},x[i,j]]<=Capacity[i]],  

     Foreach[{j,Markets}, Sum[{i,Plants},x[i,j]]>=Demand[j]]  

  ],  

 

  Goals[Minimize[TotalCost]]  

]  

This can be used from C# as follows: 

        ///  <summary> 
        ///  Holds the OML model  
        ///  </summary> 
        string  strModel = @"Model[ 
              Parameters[Sets,Plants,Markets],  
              Parameters[Reals,Capacity[Plants],Demand[Markets],Cost[Plants,Markets]],  
 
              Decisions[Reals[0,Infinity],x[Plants,Markets],TotalCost],  
 
              Constraints[  
                 TotalCost == Sum[{i,Plants},{j,Markets},Cost[i,j]*x[i,j]],  
                 Foreach[{i,Plants}, Sum[{j,Markets},x[i,j]]<=Capacity[i]],  
                 Foreach[{j,Markets}, Sum[{i,Plants},x[i,j]]>=Demand[j]]  
              ],  
 
              Goals[Minimize[TotalCost]]  
           ]" ;  
 



 

        SolverContext  context;  

        context.LoadModel( FileFormat .OML, new StringReader (strModel));  

        Solution  solution = context.Solve();  

        Console .Write( "{0}" , solution.GetReport());  

The real issue is how to handle the data. For this example we stored the data in an Access database as follows: 

  

We will use the tables Capacity and Demand and the Query Cost.  The data looks like: 

 

 

The reason to choose Access is that Access is simple database. Once we have our code working with Access, we 

can be reasonable sure that handling other database systems is easy. Essentially we are aiming to handle the least 

common denominator. The code to handle this can be as follows: 

        ///  <summary> 
        ///  Solve the problem  
        ///  </summary> 
        public  void  Solve()  
        {  
            context.LoadModel( FileFormat .OML, new StringReader (strModel));  

 
           conn = new OleDbConnection (connection);  

file:///C:/Users/erwin/AppData/Local/Temp/WindowsLiveWriter1286139640/supfiles23F3E9FD/image3.png

            foreach  ( Parameter  p in  context.CurrentModel.Parameters)  
            {  
                switch  (p.Name)  
                {  
                    case "Capacity" :  
                        setBinding(p, "select plant,capacity from capacity" ,  
                            "capacity" , new string []{ "plant" });  
                        break ;  
                    case "Demand":  
                        setBinding(p, "select market,demand from demand" ,  
                            "demand", new string []{ "market" });  
                        break ;  
                    case "Cost" :  
                        setBinding(p, "select plant,market,cost from cost" ,  
                            "cost" , new string []{ "plant" , "market" });  
                        break ;  
   
                }  
 
            }  
 
            Solution  solution = context.Solve();  
            Console .Write( "{0}" , solution.GetReport());  
 
        }  
 

 

In each binding operation we specify: 

1. The SFS parameter, which we retrieve from the CurrentModel 

2. The query to be used against the database 

3. The name of the data column 

4. The names of the index columns (passed on as an array of strings) 

The complete model looks like: 

using  System;  
using  System.Collections .Generic;  
using  System.Linq;  
using  System.Data;  
using  System.Data.OleDb;  
using  System.Data.Linq;  
using  System.Text;  
using  Microsoft.SolverFoundation.Services;  
using  System.IO;  
 
namespace OML1 
{  
    class  Trnsport  
    {  
        ///  <summary> 
        ///  Called by the OS  
        ///  </summary> 
        ///  <param name="args"></param>  
        static  void  Main( string [] args)  
        {  
            Trnsport  t = new Trnsport ();  



 

            t.Solve();  
        }  
 
        ///  <summary> 
        ///  Holds the OML model  
        ///  </summary> 
        string  strModel = @"Model[ 
              Parameters[Sets,Plants,Markets],  
              Parameters[Reals,Capacity[Plants],Demand[Markets],Cost[Plants,Markets]],  
 
              Decisions[Reals[0,Infinity],x[Plants,Markets],TotalCost],  
 
              Constraints[  
                 TotalCost == Sum[{i,Plants},{j,Markets},Cost[i,j]*x[i,j]],  
                 Foreach[{i,Plants}, Sum[{j,Markets},x[i,j]]<=Capacity[i]],  
                 Foreach[{j,Markets}, Sum[{i,Plants},x[i,j]]>=Demand[j]]  
              ],  
 
              Goals[Minimize[TotalCost]]  
           ]" ;  
 
        ///  <summary> 
        ///  Connection string for MS Access  
        ///  Use x86 architecture!  
        ///  </summary> 
        string  connection = @"Provider=Microsoft.ACE.OLEDB.12.0;Data 
Source=C: \ projects \ ms\ OML1\ OML1\ trnsport.accdb;Persist Security Info=False;" ;  
 
        ///  <summary> 
        ///   SFS 
        ///  </summary> 
        SolverContext  context;  
 

        ///  <summary> 
        ///  One db connection  
        ///  </summary> 
        OleDbConnection  conn;  
 
        ///  <summary> 
        ///   Constructor  
        ///  </summary> 
        public  Trnsport()  
        {  
            context = SolverContext .GetContext();  
        }  
 
        ///  <summary> 
        ///  get query result as DataSet  
        ///  </summary> 
        ///  <param name="connection"> connection string </param> 
        ///  <param name="query"> query as string </param> 
        ///  <returns>< /returns>  
        private  DataSet  SelectOleDbSrvRows( string  connection, string  query)  
        {  
            DataSet  ds = new DataSet ();  
            OleDbDataAdapter  adapter = new OleDbDataAdapter ();  
            adapter.SelectCommand = new OleDbCommand(query, conn);  
            adapter.Fill(ds);  
            return  ds;  
        }  
 



 

        ///  <summary> 
        ///  Perform some magic to make sure the query output arrives in OML model.  
        ///  </summary> 
        ///  <param name="p"> OML/SFS parameter</param>  
        ///  <param name="query"> database query </param> 
        ///  <param name="valueColumn"> column with values </param> 
        ///  <param name="IndexColumns"> columns with indices </param> 
        private  void  setBinding( Parameter  p, string  query, string  valueColumn, string [] 
IndexColumns)  
        {  
             DataSet  ds = SelectOleDbSrvRows(connection, query);  
             DataTable  dt = ds.Tables[0];  
             p.SetBinding(dt.AsEnumerable(), valueColumn, IndexColumns);  
        }  
 
        ///  <summary> 
        ///  Solve the problem  
        ///  </summary> 
        public  void  Solve()  
        {  
            context.LoadModel( FileFormat .OML, new StringReader (strModel));  
 
            conn = new OleDbConnection (connection);  
 
            foreach  ( Parameter  p in  context.CurrentModel.Parameters)  
            {  
                switch  (p.Name)  
                {  
                    case "Capacity" :  
                        setBinding(p, "select plant,capacity from capacity" ,  
                            "c apacity" , new string []{ "plant" });  
                        break ;  
                    case "Demand":  
                        setBinding(p, "select market,demand from demand" ,  
                            "demand", new string []{ "market" });  
                        break ;  
                    case "Cost" :  
                        setBinding(p, "select plant,market,cost from cost" ,  
                            "cost" , new string []{ "plant" , "market" });  
                        break ;  
   
                }  
 
            }  
 
            Solution  solution = context.Solve();  
            Console .Write( "{0}" , solution.GetReport());  
 
        }  
 
    }  
}  
 

4.2 CALLING C# FROM EXCEL 

Sometimes models are not suited to be expressed in OML only. An example would be because we need to solve 

different models or need to implement an algorithm. In this case we really need to use the Solver Foundation 

!tLΩǎΦ /ŀƴ ǿŜ ǎǘƛƭƭ ǳǎŜ 9ȄŎŜƭ ŀǎ ŦǊƻƴǘ-end to host the user-interface? The answer is yes. There are basically two ways 



 

to call .Net in general or C# specifically from Excel. The first one is to create a .Net DLL with appropriate COM 

interfaces. Then this DLL can be called from VBA. This approach is illustrated in section Error! Reference source not 

found.. Another approach is to use VSTO: a framework to call .Net from Office applications. This is demonstrated 

in section 5.13. 

5 EXAMPLES 

In this section we will implement a few examples. The first example was a simple linear programming problem 

based on a transportation problem and was discussed in the introduction section. Here we discuss a few other 

models. 

The example models are not real-life models. We use here smaller artificial models, as they can be explained easier 

while they can illustrate a certain modeling issue in a more succinct way. Real-life models tend to be large, 

complicated, and messy and they have lots of extra details that just obscure the issues we want to demonstrate. 

5.1 A DIET PROBLEM 

The diet problem goes back to work by the economist George Stigler (Stigler, 1945), although earlier formulations 

have been mentioned (Murphy, 1996). It is one of the first optimƛȊŀǘƛƻƴ ǇǊƻōƭŜƳǎ ǘƻ ōŜ ǎǘǳŘƛŜŘ ōŀŎƪ ƛƴ ǘƘŜ мфолΩǎ 

ŀƴŘ плΩǎΦ Lǘ ǿŀǎ ŦƛǊǎǘ ƳƻǘƛǾŀǘŜŘ ōȅ ǘƘŜ !ǊƳȅΩǎ ŘŜǎƛǊŜ ǘƻ ƳŜŜǘ ǘƘŜ ƴǳǘǊƛǘƛƻƴŀƭ ǊŜǉǳƛǊŜƳŜƴǘǎ ƻŦ ǘƘŜ ŦƛŜƭŘ DLΩǎ ǿƘƛƭŜ 

minimizing the cost. 

Below we show a part of the table that gives nutrient contents of different commodities per dollar spent. 

id description units price weight calories protein calcium iron vitamin A 

   
aug 15 1939 edible per $1 

    

   
(cents) (grams) (1000) (grams) (grams) (mg.) (1000 IU) 

flour        Wheat Flour (Enriched) 10 lb. 36 12600 44.7 1411 2 365        

macaroni     Macaroni 1 lb. 14.1 3217 11.6 418 0.7 54        

cereal       Wheat Cereal (Enriched) 28 oz. 24.2 3280 11.8 377 14.4 175        

cornflakes   Corn Flakes 8 oz. 7.1 3194 11.4 252 0.1 56        

cornmeal     Corn Meal 1 lb. 4.6 9861 36 897 1.7 99 30.9 

grits        Hominy Grits 24 oz. 8.5 8005 28.6 680 0.8 80        

rice         Rice 1 lb. 7.5 6048 21.2 460 0.6 41        

In addition we have a table of minimum requirements per day: 

recommended daily allowances for a moderately active man 

      calories     3 

    protein      70 

    calcium      0.8 

    iron         12 

    vitaminA     5 

    thiamine     1.8 

    riboflavin   2.7 

    



 

niacin       18 

    ascorbicAcid 75 

    
Together we can compute the least expensive diet, to keep the poor soul subject to this diet-plan alive. The linear 

programming model is fairly straightforward: 

min ὄόώὧ
ὧɴὅ

ὅέὲὸὩὲὸὧ,ὲὄόώὧ ὃὰὰέύὥὲὧὩὲ  ᶅ ὲᶰὔ
ὧɴὅ

 

This is very simple: 

Model[  
   Parameters[Sets,C,N] ,  
   Parameters[Reals,content[C,N],allowance[N]],  
    
   Decisions[Reals[0,Infinity],TotalCost,Buy[C]],  
   Constraints[  
     TotalCost == Sum[{c,C},Buy[c]],  
     Foreach[{n,N}, Sum[{c,C}, content[c,n]*Buy[c]] >= allowance[n]]  
   ],  
   Goals[Minimize[TotalCost]]  
]  
 

The objective is very simple as the variable Buy is expressed in dollars. If the unit was related to volume we would 

ƘŀǾŜ ǎŜŜƴ ŀ Ŏƻǎǘ ŎƻŜŦŦƛŎƛŜƴǘ ŘƛŦŦŜǊŜƴǘ ŦǊƻƳ ŀƭƭ ƻƴŜΩǎΦ ²Ŝ ƘŀǾŜ ǇƭŀŎŜŘ ǘƘŜ ƻōƧŜŎǘƛǾŜ ƛƴ ǘƘŜ /ƻƴǎǘǊŀƛƴǘǎ ǎŜŎǘƛƻƴΦ Lǘ ƛǎ 

also possible to write: 

Model[  
   Parameters[Sets,C,N],  
   Parameters[Reals,content[C,N],allowance[N]],  
    
   Decisions[Reals[0,Infinity],Buy[C]],  
   Constraints[  
     Foreach[{n,N}, Sum[{c,C}, content[c,n]*Buy[c]] >= allowance[n]]  
   ],  
   Goals[Minimize[TotalCost - > Sum[{c,C},Buy[c]]]]  
]  
 

 Here we have a real objective function Sum[{c,C},Buy[c]]  with a label TotalCost .  

This compares directly to the original GAMS model: 

positive  variable  x(c)  'dollars of food to be purchased daily   (dollars)'  

;  

 

free  variable  cost      'total food bill                         (dollars)'  

 

equations  

    nb(n)    'nutrient balance  (units)'  

    cb       'cost balance      (dollars)'  



 

;  

 

nb(n).. sum(c, data(c,n)*x(c)) =g= allowance(n);  

cb..    cost=e= sum(c, x(c));  

 

model  diet 'stiglers diet problem' /  nb,cb / ;  

solve  diet minimizing cost using lp;  

 

The problem is however with the data. A missing entry for a Content[c,n]  means it has zero content for that 

particular nutrient. In GAMS this is handled automatically as its sparse matrix storage implies that non-existent and 

zero is the same thing. 

To be able to import the data in OML and sets we need to reorganize the data substantially. I use the table format: 

C N Content 

Flour Calories 44.7 

Flour Protein 1411 

Flour Calcium 2 

Flour Iron 365 

Flour vitaminA 0 

Flour Thiamine 55.4 

Flour Riboflavin 33.3 

Flour Niacin 441 

Flour ascorbicAcid 0 

Macaroni Calories 11.6 

Macaroni Protein 418 

Macaroni Calcium 0.7 

Macaroni Iron 54 

Macaroni vitaminA 0 

Macaroni Thiamine 3.2 

Macaroni Riboflavin 1.9 

Macaroni Niacin 68 

Macaroni ascorbicAcid 0 

Here we introduced many zeros where needed as OML does not handle sparse tables. Some of the inserted 

records are highlighted in the above table. We also removed some records (price, weight) that were not required.  

For larger data sets adding the zero records explicitly may be a lot of work and it may use up a lot of space in the 

spreadsheet. 

In general we need to be very precise with the data to match exactly the parameters and sets in the model. We 

cannot repair any problems with the data in the OML model. This is very different from GAMS where one often 

takes the data as is, and then manipulate the data until suitable for use in the model equations. 

5.2 MAX FLOW, A NETWORK MODEL 

The max flow problem can be stated as: 



 

maxὪ

Ὥᶅ: ὼὮ,Ὥ
ὥὶὧ(Ὦ,Ὥ)

ὼὭ,Ὦ
ὥὶὧ(Ὥ,Ὦ)

=
Ὢ  if i is a source
Ὢ   if i is a sink
0   otherwise

0 ὼὭ,Ὦ ὧὥὴ(Ὥ,Ὦ)

 

The model is not too difficult to state in OML. An issue is that many networks are sparse, a concept not really 

supported by OML.  The GAMS formulation can look like: 

$ontext  

 

   max flow network example  

 

   Data from example in  

     Mitsuo Gen, Runwei Cheng, Lin Lin  

     Network Models and Optimization: Multiobjective Genetic Algorithm Approach  

     Springer, 2008  

 

   Erwin Kalvelagen, Amsterdam Optimization, May 2008  

 

$offtext  

 

 

sets  

   i 'nodes' /node1*node11/  

   source(i)   /node1/  

   sink(i)    /node11/  

;  

 

alias (i,j);  

 

parameter  capacity(i,j) /  

   node1.node2   60  

   node1.node3   60  

   node1.node4   60  

   node2.node3   30  

   node2.n ode5   40  

   node2.node6   30  

   node3.node4   30  

   node3.node6   50  

   node3.node7   30  

   node4.node7   40  

   node5.node8   60  

   node6.node5   20  

   node6.node8   30  

   node6.node9   40  

   node6.node10  30  

   node7.node6   20  

   node7.node10  40  

   node8.node9   30  

   node8.node11  60  

   node9.node10  30  

   node9.node11  50  

   node10.node11 50  

/ ;  

 

 

 

set  arcs(i,j);  

arcs(i,j)$capacity(i,j) = yes ;  

display  arcs;  



 

 

parameter  rhs(i);  

rhs(source) = - 1;  

rhs(sink) = 1;  

 

variables  

   x(i,j) 'flow along arcs'  

   f      'total flow'  

;  

 

positive  variables  x;  

x.up(i,j) = capacity(i,j);  

 

equations  

   flowbal(i)  'flow balance'  

;  

 

flowbal(i)..   sum(arcs(j,i), x(j,i)) -  sum(arcs(i,j), x(i,j)) =e= f*rhs(i);  

 

model  m/flowbal/ ;  

 

solve  m maximizing f using lp;  

 

We discussed this already in section 3.4.7. The sparse network cannot be handled directly by OML. Of course we 

could generate a dense graph by introducing capacities for all arcs between any two nodes. For larger networks 

this is not a reasonable approach. As a workaround, we number the arcs and have as data From[.] and To[.]. This 

looks like: 

   

The complete model can look like: 



 

Model[  
  
  Parameters[Sets,Arcs,Node s],  
  Parameters[Reals,Capacity[Arcs]],  
  Parameters[Integers,Rhs[Nodes],From[Arcs],To[Arcs],ArcNum[Arcs],NodeNum[Nodes]],  
   
  Decisions[Reals[0,Infinity],maxflow,flow[Arcs]],  
 
  Constraints[  
 
   Foreach[{j,Nodes},  
 FilteredSum[{a,Arcs},To[a]==NodeNum[j],flow[a]] -   
         FilteredSum[{a,Arcs},From[a]==NodeNum[j],flow[a]] == maxflow*Rhs[j]],  
 
   Foreach[{a,Arcs},flow[a]<=Capacity[a]]  
],  
 
   Goals[Maximize[maxflow]]  
 
]  

 

 



 

 

The node balance equation becomes a little bit unwieldy and less readable with this approach. This is a small, 

artificial example but note that quite a few models have some network component. It illustrates that leaving out 

sparse data handling and multidimensional sets to simplify a modeling language, although not a show stopper, 

comes with a cost in the form of additional complexity for the modeler. Note that in this model we still have a 

possibly large node table to maintain (the GAMS model handles this sparse). 

5.3 THE SOCIAL GOLFER PROBLEM 

I am recently involved in a practical application of a scheduling problem related to the Social Golfer Problem. The 
particular application is somewhat more complicated, but we could use the GAMS/Cplex model described here as 
a starting point. 

The problem is to find a good schedule for a number (N) of golf players. They play T rounds in groups of size GS. I.e. 
we have T*GS=N. The schedule has to be designed that each golfer meets another golfer at most one time. 

For smaller instances of the pure Social Golfer Problem, it is convenient to use a CSP approach:  

http://www.cs.brown.edu/~sello/golf.html
http://yetanothermathprogrammingconsultant.blogspot.com/2008/06/variant-of-social-golfer-problem.html


 

variable ὼὭ,Ὣ,ὸ {ɴ0,1} Binary variable indicating if player i is playing in group g in round t. 

ὼὭ,Ὣ,ὸ= 1  ᶅ Ὥ,ὸ

Ὣ

 A player has to play each round in exactly one group. 

ὼὭ,Ὣ,ὸ= ὋὛ  ᶅ Ὣ,ὸ

Ὥ

 Each group consist of GS players 

ὼὭ,Ὣ,ὸὼὮ,Ὣ,ὸ 1  ᶅ Ὥ,Ὦ

Ὣ,ὸ

 Restrict the number of times players i and j meet. This is a non-linear constraint, 
but that is no problem in a CSP setting. The equation is specified here for each 
combination i and j. Note however that if we compared i and j we no longer have to 
inspect j and i, so we can restrict the number of equations by exploiting symmetry 
here. 

Without loss of generality we can fix the first round and fix the first player. Here is a formulation in OML: 

Model[  
 
// N : number of golfers  
// NG : number of groups  
// T :  number of rounds  
// GS : group size (N/NG)  
 
 Parameters[Integers,N=16,NG=4,T=5,GS=4],  
 // Would prefer: GS=N/NG but OML does not allow (constant) expressions  
 // in parameter statements  
 
 Decisions[  
  Integers[0,1],  
   Foreach[{i,N},{g,NG},{t,T},x[i,g,t]]  
 ],  
 
 
 Constraints[  
     // each golfer has to play each round  
     Foreach[{i,N},{t,T},Sum[{g,NG},x[i,g,t]] == 1],  
 
     // form groups  
     Foreach[{g,NG},{t,T},Sum[{i,N},x[i,g,t]] == GS],  
 
     // golfer i and j meet at most one time  
     // Foreach[{i ,N}, FilteredForeach[{j,N},i!=j, Sum[{g,NG},{t,T},x[i,g,t]*x[j,g,t]] <= 1]],  
     // We exploit symmetry here:  
     Foreach[{i,N}, Foreach[{j,i+1,N}, Sum[{g,NG},{t,T},x[i,g,t]*x[j,g,t]] <= 1]],  
 
 
     // fix first round  
     Foreach[{g,NG},{k,GS},x[GS*g+k, g,0]==1],  
 
     // fix first golfer  
     Foreach[{t,1,T},x[0,0,t]==1]  
 ]  
 
]  



 

The resulting scheduling looks like: 

        

  

Schedule 
 

        
        
  

round 1 round 2 round 3  round 4 round 5 

 
 

group 1 

Golfer 0 Golfer 0 Golfer 0 Golfer 0 Golfer 0 

 
 

Golfer 1 Golfer 6 Golfer 4 Golfer 5 Golfer 7 

 
 

Golfer 2 Golfer 9 Golfer 8 Golfer 11 Golfer 10 

 
 

Golfer 3 Golfer 13 Golfer 12 Golfer 14 Golfer 15 

 

 
group 2 

Golfer 4 Golfer 3 Golfer 3 Golfer 2 Golfer 2 

 
 

Golfer 5 Golfer 7 Golfer 5 Golfer 6 Golfer 4 

 
 

Golfer 6 Golfer 8 Golfer 9 Golfer 8 Golfer 9 

 
 

Golfer 7 Golfer 14 Golfer 15 Golfer 15 Golfer 14 

 

 
group 3 

Golfer 8 Golfer 2 Golfer 2 Golfer 3 Golfer 3 

 
 

Golfer 9 Golfer 5 Golfer 7 Golfer 4 Golfer 6 

 
 

Golfer 10 Golfer 10 Golfer 11 Golfer 10 Golfer 11 

 
 

Golfer 11 Golfer 12 Golfer 13 Golfer 13 Golfer 12 

 

 
group 4 

Golfer 12 Golfer 1 Golfer 1 Golfer 1 Golfer 1 

 
 

Golfer 13 Golfer 4 Golfer 6 Golfer 7 Golfer 5 

 
 

Golfer 14 Golfer 11 Golfer 10 Golfer 9 Golfer 8 

 
 

Golfer 15 Golfer 15 Golfer 14 Golfer 12 Golfer 13 

 

  
  

  
  

  
        

Indeed the meet counts are what we want: 

Meet count  

                 

                 

 
golfer 0 golfer 1 golfer 2 golfer 3 golfer 4 golfer 5 golfer 6 golfer 7 golfer 8 golfer 9 golfer 10 golfer 11 golfer 12 golfer 13 golfer 14 golfer 15 

golfer 0 
 

1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 

golfer 1 1 
 

1 1 1 1 1 1 1 1 1 1 1 1 1 1 

golfer 2 1 1 
 

1 1 1 1 1 1 1 1 1 1 1 1 1 

golfer 3 1 1 1 
 

1 1 1 1 1 1 1 1 1 1 1 1 

golfer 4 1 1 1 1 
 

1 1 1 1 1 1 1 1 1 1 1 

golfer 5 1 1 1 1 1 
 

1 1 1 1 1 1 1 1 1 1 

golfer 6 1 1 1 1 1 1 
 

1 1 1 1 1 1 1 1 1 

golfer 7 1 1 1 1 1 1 1 
 

1 1 1 1 1 1 1 1 

golfer 8 1 1 1 1 1 1 1 1 
 

1 1 1 1 1 1 1 

golfer 9 1 1 1 1 1 1 1 1 1 
 

1 1 1 1 1 1 

golfer 10 1 1 1 1 1 1 1 1 1 1 
 

1 1 1 1 1 

golfer 11 1 1 1 1 1 1 1 1 1 1 1 
 

1 1 1 1 

golfer 12 1 1 1 1 1 1 1 1 1 1 1 1 
 

1 1 1 

golfer 13 1 1 1 1 1 1 1 1 1 1 1 1 1 
 

1 1 

golfer 14 1 1 1 1 1 1 1 1 1 1 1 1 1 1 
 

1 

golfer 15 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 
 

 

A different formulation would be to use an integer variable x[i,t]=g which corresponds to x[i,g,t]=1 in the above 
model.  

variable ὼὭ,ὸ {ɴ1,ȣ,ὔὋ} Integer variable indicating in which group golfer i plays in round t. 



 

1 = ὋὛ

Ὥ|ὼὭ,ὸ=Ὣ

  ὸᶅ,Ὣ The number of golfers in a group must be NG 

1 1  ᶅ Ὥ,Ὦ

ὸ|ὼὭ,ὸ=ὼὮ,ὸ

 The number of times golfer i and j meet. Again we can exploit symmetry here. 

The OML representation can look like:  

Model[ 
 
// N : number of golfers 
// NG : number of groups 
// T : number of rounds 
// GS : group size (N/NG) 
 
  Parameters[Integers,N=16,NG=4,T=5,GS=4], 
  // Parameters[Integers,N=32,NG=8,T=10,GS=4], 
 
  Decisions[ 
    // Integers[0,NG-1], 
      Integers[0,3], 
       Foreach[{i,N},{t,T},x[i,t]] 
  ], 
 
  Constraints[ 
     // form groups 
     Foreach[{g,NG},{t,T},Sum[{i,N},AsInt[x[i,t]==g]] == GS], 
 
     // golfer i and j meet at most one time 
     Foreach[{i,N}, {j,i+1,N}, Sum[{t,T},AsInt[x[i,t]==x[j,t]]] <= 1], 
 
     // fix first round 
     Foreach[{g,NG},{k,GS},x[GS*g+k,0]==g], 
 
     // fix first golfer 
     Foreach[{t,1,T},x[0,t]==0] 
  ] 
] 

This formulation would not be possible with a MIP solver. The performance of the first formulation seems a little 
bit better. 

In a MIP formulation we can go back to our binary variables x[i,g,t]. The binary multiplication in the meet count 

equation can be linearized as: 

άὭ,Ὦ,Ὣ,ὸ ὼὭ,Ὣ,ὸ

άὭ,Ὦ,Ὣ,ὸ ὼὮ,Ὣ,ὸ

άὭ,Ὦ,Ὣ,ὸ ὼὭ,Ὣ,ὸ+ ὼὮ,Ὣ,ὸ 1

άὭ,Ὦ,Ὣ,ὸ 1

Ὣ,ὸ

 



 

where 0 άὭ,Ὦ,Ὣ,ὸ 1is a new variable; this variable can be continuous. The first two inequalities can actually be 

droppedΣ ŀǎ ǿŜ ŀǊŜ ƻƴƭȅ ƛƴǘŜǊŜǎǘŜŘ ƛƴ ƪŜŜǇƛƴƎ ǘƘŜ ƴǳƳōŜǊ ƻŦ ƳΩǎ ǘƘŀǘ ŀǊŜ ƻƴŜ ŘƻǿƴΦ ²Ŝ ǎŜŜ ǘƘŀǘ ŀ aLt 

formulation needs many more equations and variables than a corresponding CSP model. 

5.4 JOB SHOP SCHEDULING 

Scheduling models are sometimes very difficult to solve as a mathematical programming problems. A good 

example of this is the standard Job Shop Scheduling problem.  

Probably the best way to explain the problem is looking at a data set: 

 
task1 task2 task3 task4 task5 task6 

 
machine time machine time Machine time machine time machine time machine time 

job1 m2 1 m0 3 m1 6 m3 7 m5 3 m4 6 

job2 m1 8 m2 5 m4 10 m5 10 m0 10 m3 4 

job3 m2 5 m3 4 m5 8 m0 9 m1 1 m4 7 

job4 m1 5 m0 5 m2 5 m3 3 m4 8 m5 9 

job5 m2 9 m1 3 m4 5 m5 4 m0 3 m3 1 

job6 m1 3 m3 3 m5 9 m0 10 m4 4 m2 1 

Each job has to go through a number of stages (called tasks here) on different machines. The times a task occupies 

a machine is listed in the table. Each machine can only work on one task at the time. The goal is to design a 

schedule that minimizes the total make span, i.e. the time that the last task is finished. 

The basic model looks like: 

variable ὼὮ,ά 0 Start time of task of running job j on machine m 

variable ώὮ,Ὧ,ά {ɴ0,1} A binary variable indicating whether job j comes after job k on machine m  

ὼὮ,ά2 ὼὮ,ά1 + ὸὮ,ά1 Precedence equations. For certain combinations (j,m1,m2) we need to 

prescribe a sequencing: first execute task on machine m before we can 

execute on machine m2. The sequencing data is taken from the table above. 

E.g. ὼὮέὦ1,άὥὧὬὭὲὩ0 ὼὮέὦ1,άὥὧὬὭὲὩ2 + 1. We only need to do this for tasks 

that immediately follow each other. This is not very easy to do in OML as we 

ŘƻƴΩǘ ƘŀǾŜ ǎǇŀǊǎŜ Ƴǳƭǘƛ-dimensional sets. 

ὼὮ,ά ὼὯ,ά + ὸὯ,ά ὓώὮ,Ὧ,ά

ὼὯ,ά ὼὮ,ά + ὸὮ,ά ὓ(1 ώὮ,Ὧ,ά)
 

No overlap equations. These equations make sure a machine is occupied by 

ƻƴƭȅ ǳǇ ǘƻ ƻƴŜ Ƨƻō ŀǘ ǘƘŜ ǘƛƳŜΦ ¢Ƙƛǎ ƛǎ ƳƻŘŜƭŜŘ ōȅ ŀƴ ΨƻǊΩ ŎƻƴŘƛǘƛƻƴΥ Ƨƻō j is 

executed on machine m before or after job k. This is a big-M formulation: we 

need to find an appropriate value for it. In the model we just use the sum of 

all processing times for this: no job will be scheduled later than that. This 

type of constraint is typical in scheduling applications. 



 

minimize ᾀ
ᾀ ὼὮ,ά + ὸὮ,ά

 
The objective is to minimize the total make span. This is modeled by 

minimizing the finishing time of the last task.  

This formulation is from (Manne, 1960). The OML representation looks like: 

// Job shop scheduling  
Model[  
 
  Parameters[Sets,Job,Machine,Task],  
  Parameters[Reals,Time[Job, Machine],TotTime[]],  
  Parameters[Integers,MachNo[Machine],Mach1[Task,Job],Mach2[Task,Job],JobNo[Job]],  
   
  Decisions[Reals[0,Infinity],  
         x[Job,Machine],  // start time of sub - task  
         MakeSpan           // total make span of the problem  
  ],  
  // the 0 - 1 variables deal with overlap  
  Decisions[Integers[0,1],y[Job,Job,Machine]],  
 
  Constraints[  
 
    // precedence  
     Foreach[{t,Task},{j,Job},  
         FilteredSum[{m2,Machine}, MachNo[m2]==Mach2[t,j],x[j,m2]] >=  
         FilteredSum[{m1,Machine},MachNo[m1]==Mach1[t,j],x[j,m1] + Time[j,m1]]  
     ],  
 
     // no overlap  
     Foreach[{m,Machine},{j,Job},  
        FilteredForeach[{k,Job},JobNo[j]<JobNo[k],  
           x[j,m] >= x[k,m] +  Time[k,m] -  TotTime[]*y[j,k,m]  
       ]],          
 
     Foreach[{m,Machine},{j,Job},  
        FilteredForeach[{k,Job},JobNo[j]<JobNo[k],  
           x[k,m] >= x[j,m] + Time[j,m] -  TotTime[]*(1 - y[j,k,m])  
       ]],  
 
      // make span  
      Foreach[{j,Job},{m,Machine},  
         MakeSpan >= x[j,m] + Time[j,m]  
      ]  
  ],  
 
   Goals[Minimize[makespan - >MakeSpan]]  
  
]  

The precedence equations are somewhat complicated as we need to simulate a sparse set here: 

ὼὮ,ά2 ὼὮ,ά1 + ὸὮ,ά1  ᶅ (Ὦ,ά1,ά2) ᶰὛ 

The data as presented in the table is not suited to be imported directly into the model. In a different sheet we 

prepare the data ready for consumption by the model: 



 

 

This problem with six jobs and six machines solves quickly with the Gurobi mip solver. The model can be simplified 

ǎƻƳŜǿƘŀǘ ǿƘŜƴ ǳǎƛƴƎ /{t ƳƻŘŜƭƛƴƎΥ ǘƘŜ ΨƻǊΩ ŎƻƴŘƛǘƛƻƴ ōŜŎƻƳŜǎ Ŝŀǎȅ ŀƴŘ ǿŜ Ŏŀƴ ŘǊƻǇ ǘƘŜ ōƛƴŀǊȅ y variables. 

However the CSP model did not solve as quickly as the MIP model. 

With some VBA code it is easy to create a GANTT chart of the solution: 



 

 

This shows how jobs are scheduled. A different view would be: 

 

Larger instances can be difficult to solve. A famous benchmark model called ft10 (Fisher & Thompson, 1963) with 

10 jobs and 10 machines was only solved to optimality in (Carlier & Pinson, 1989) after 25 years being unsolved 

(Jain & Meeran, 1998). Nowadays we can solve this problem using a MIP solver like Gurobi in less than five minutes 

on a standard PC. 

The data for ft10 looks like: 

 
task1 task2 task3 task4 task5 task6 task7 task8 task9 task10 

 
machine time machine time machine time machine time machine time machine time machine time machine time machine time machine time 

job1 m0 29 m1 78 m2 9 m3 36 m4 49 m5 11 m6 62 m7 56 m8 44 m9 21 

job2 m0 43 m2 90 m4 75 m9 11 m3 69 m1 28 m6 46 m5 46 m7 72 m8 30 

job3 m1 91 m0 85 m3 39 m2 74 m8 90 m5 10 m7 12 m6 89 m9 45 m4 33 

job4 m1 81 m2 95 m0 71 m4 99 m6 9 m8 52 m7 85 m3 98 m9 22 m5 43 

job5 m2 14 m0 6 m1 22 m5 61 m3 26 m4 69 m8 21 m7 49 m9 72 m6 53 

job6 m2 84 m1 2 m5 52 m3 95 m8 48 m9 72 m0 47 m6 65 m4 6 m7 25 

job7 m1 46 m0 37 m3 61 m2 13 m6 32 m5 21 m9 32 m8 89 m7 30 m4 55 

job8 m2 31 m0 86 m1 46 m5 74 m4 32 m6 88 m8 19 m9 48 m7 36 m3 79 

job9 m0 76 m1 69 m3 76 m5 51 m2 85 m9 11 m6 40 m7 89 m4 26 m8 74 

job10 m1 85 m0 13 m2 61 m6 7 m8 64 m9 76 m5 47 m3 52 m4 90 m7 45 

The results: 



 

 

5.5 MAGIC SQUARES 

A magic square is a square (n × n) matrix of unique integers 1,..,n² such that the row sums, column sums and sums 

over the two diagonals all yield the same number. For a detailed explanation see 

http://en.wikipedia.org/wiki/Magic_square.  

The actual value of the sum over the rows, columns and diagonals can be easily derived. The sum over all cells is 

Ὥ=
ὲ2

2
(ὲ2 + 1)

ὲ2

Ὥ= 1

 

Each row sum is equal, so a row sum is this number divided by n. As a result the row, column and diagonal sums 

are equal to 

1

2
ὲ(ὲ2 + 1) 

http://en.wikipedia.org/wiki/Magic_square


 

Finding a magic square of size n is a feasibility problem (there is no objective: just find a feasible solution). The 

problem is quite suited for the CSP solver, as we can use the all-different constraint to create a more efficient 

model than is possible with a MIP formulation. The basic model can look like: 

Model[  
 
  Parameters[Integers,N[]],  
  Decisions[Integers[1,Infinit y],Foreach[{i,1,N[]+1},{j,1,N[]+1},x[i,j]]],  
 
  Constraints[  
 
// bounds on variables:  
     Foreach[{i,1,N[]+1},{j,1,N[]+1},1 <= x[i,j]<= N[]^2],  
 
// row and column sums:  
     Foreach[{i,1,N[]+1}, Sum[{j,1,N[]+1},x[i,j]] == N[]*(N[]^2+1)/2],  
     Foreach[{j,1,N[]+1}, Sum[{i,1,N[]+1},x[i,j]] == N[]*(N[]^2+1)/2],      
 
// diagonals:  
     Sum[{i,1,N[]+1},x[i,i]] == N[]*(N[]^2+1)/2,  
     Sum[{i,1,N[]+1},x[i,N[] - i+1]] == N[]*(N[]^2+1)/2,  
 
// additional constraint: sum of all cells  
     Sum[{i,1,N[] +1},{j,1,N[]+1},x[i,j]] == N[]^2*(N[]^2+1)/2,  
 
// symmetry breaking constraints  
     x[1,1] <= x[1,N[]] - 1,  
     x[1,1] <= x[N[],N[]] - 1,  
     x[1,1] <= x[N[],1] - 1,  
     x[1,N[]] <= x[N[],1] - 1,  
 
// all different  
     Unequal[Foreach[{i,1,N[]+1},{j,1,N[]+1},x [i,j]]]  
   ]  
 
]  

 

There are a number of points we can make about this model. 

a.  We repeat the expression N[]*(N[]^2+1)/2  several times. That is really considered bad modeling. 

Unfortunately OML does not allow to say Parameters[Integers,M=N[]*(N[]^2+1)/2 ] . We could have 

calculated this number in Excel and import that, but that violates another rule: keep interfaces as small as 

possible. 

b. We use here indexing 1..N. Although most modelers would prefer to use X[1..n,1..n] , OML models 

are often better readable when using X[0..n - 1,0..n - 1] . The reason is that Sum[{i,N},..]  and 

&ÏÒÅÁÃÈǁǅÉƗ.ǆƗƛǂ actually means loop over i=0..N-1. A loop i=1..N can be written as 

Sum[{i,1,N+1},..]  and &ÏÒÅÁÃÈǁǅÉƗʦƗ.˩ʦǆƗƛǂ. The visual clutter is exacerbated by using the N[]  

notation. My guess is that this zero-based indexing is chosen as MSF is really programmer-centric: it 

primarily targets and is developed by software developers as opposed to modelers. 

c. The all-different constraint is very useful for models like this. This construct cannot be translated 

efficiently to a MIP based model (Williams & Yan, 2001). 

d. In MSF version 1.0, a bug did not allow us to declare a variable sized using N[]. I had to introduce a 

constant MaxN and size x accordingly. This resulted in unused variables. 



 

e. In MSF version 1.0, the unused variables were printed in the solution sheet, in version 1.1, they are not. 

This caused problems with some of the spreadsheet applications I developed. It is more important to 

make new versions backward compatible if users build bigger application around a framework. 

f. We use a few simple constraints to reduce some symmetry. For more examples of symmetry breaking 

constraints see section 5.9. 

Thƛǎ ŀǇǇƭƛŎŀǘƛƻƴ Ƙŀǎ ŀ ƭƛǘǘƭŜ D¦LΦ ¢ƘŜ ŎƻŘŜ ŦƻǊ ǘƘƛǎ Ƙŀǎ ōŜŜƴ ƛƳǇƭŜƳŜƴǘŜŘ ƛƴ 9ȄŎŜƭΩǎ ±.!Φ 

 

After setting N, the boards is resized to N×N, and the solution is reset to all zeros. After pressing the Solve button, 

MSF will solve the model: 



 

 

Finally when you go back to the main page, the result is displayed on the board: 

 

The performance of this model depends on the algorithm settings. From the Summary we see that we used 

TreeSearch/DomainOverWeightedTree/SuccessPrediction. The default options cause the model to solve much 

slower. 

  

 



 

5.6 SUDOKU 

The Sudoku puzzle (http://en.wikipedia.org/wiki/Sudoku) can be conveniently coded as a CSP problem. 

Model[  
 
  Parameters[Sets,I,J],  
  Parameters[Integers,d[I,J]],  
 
  Decisions[Integers[1,9],x[I,J]],  
 
  Constraints[  
    FilteredForeach[{i,I},{j,J},d[i,j]>0,x[i,j]==d[i,j]],  
    Foreach[{i,I},Unequal[Foreach[{j,J},x[i,j]]]],  
    Foreach[{j,J},U nequal[Foreach[{i,I},x[i,j]]]],   
    Foreach[{ib,3},  
        Foreach[{jb,3},  
            Unequal[Foreach[{i,ib*3,ib*3+3},{j,jb*3,jb*3+3},x[i,j]]]  
         ]  
    ]  
]  
 
]  

Notice how Foreach is placed both inside and outside the Unequal construct. 

A small GUI is easily implemented in Excel: 

 

Checking whether the problem has a unique solution is easy: press the Next button. It should give: Final Solution 

Found. 

http://en.wikipedia.org/wiki/Sudoku


 

5.7 LANGFORD SEQUENCES 

[ŀƴƎŦƻǊŘΩǎ ǇǊƻōƭŜƳ ƛǎ ǘƻ ŦƻǊƳ ŀ ǎŜǉǳŜƴŎŜ ƻŦ ƴǳƳōŜǊǎ мΣΧΣƴ ǿƛǘƘ ǎƻƳŜ ǎǇŜŎƛŀƭ ŎƘaracteristics. E.g. if we have 4 

pairs of numbers denoted by L(2,4) then we can form:   

 

Between any pairs of value k there are k blocks with other numbers. For more information see: 

http://www.lcl ark.edu/~miller/langford.html. To model this we form variables Position[i,j] with м Җ ƛ Җ ƴ the 

numbers to use and м Җ Ƨ Җ ǎ the number of duplicates (j=2 means pairs). The constraints are simple: the position 

should be between 1 and n*s and should be unique. This can be easily modeled using the Unequal construct. 

Secondly we impose that x[i,j] - x[i,j-1] = i+1. The complete model is: 

Model[  
 
  Parameters[Integers,N[],S[] ],  
 
  Decisions[Integers,Foreach[{i,1,N[]+1},{j,1,S[]+1},position[i,j]]],    
  
 Constraints[  
   Foreach[{i,1,N[]+1},{j,1,S[]+1},  1 <= position[i,j] <= S[]*N[]],  
 
   Foreach[{i,1,N[]+1},{j,2,S[]+1},  position[i,j] == position[i,j - 1] + i + 1 ],  
   Unequal[Foreach[{i,1,N[]+1},{j,1,S[]+1},position[i,j]]]    
  ]  
 
]  

For a MIP formulation see: http://www.amsterdamoptimization.com/pdf/langford.pdf. There is more research 

available about CSP modeling with respect to this problem (Smith, 2000). 

The GUI allows you to select from a number of possible sequences: 

 

After selecting the problem, you can press the Solve button. The reported solution  looks like: 

http://www.lclark.edu/~miller/langford.html
http://www.amsterdamoptimization.com/pdf/langford.pdf


 

 

After going back to the main sheet, the solution is presented in a different, graphical format: 

  



 

VBA code was used to form this solution. 

5.8 TRAVELING SALESMAN PROBLEM 

In theory it is easy to formulate a TSP when the all-different constraint is present. Let x[i]  be the i
th
 city visited, 

then an OML-like formulation could look like: 

Model[  
  Parameters[Integers,N=14],  
  Parameters[Sets[Integers],city],  
  Parameters[Reals,dist[city,city]],  
  Decisions[Integers[0, N- 1],x[city]],  
  Constraints[  
     Unequal[Foreach[{i,city},x[i]]]  
  ],  
  Goals[  
    Minimize[Sum[{i,city}, dist[x[i],x[i++1]] ]]  
  ]  
]  
 

This does not work completely. First, bounds cannot contain a symbolic constant. So we need to write [0,13]  

instead of [0, N- 1]  as bounds:  Decisions[Integers[0, 13],x[city]].  The objective has a few more 

difficulties. There is no circular lead/lag operator in OML (like the --  or ++ operator in GAMS).  This can be solved 

by introducing a parameter next[city] . This can be calculated in Excel and imported in the OML model. 

Furthermore, we cannot use a variable as an index, so dist[x[i],x[next[i]]]  is not a valid construct 

(some languages geared towards solving CSP problems allow this; it adds concise expressiveness to the language 

that may help the modeler). The error message indicates this is not related to OML per se but rather to solver 

support. Also it is not allowed to use something like FilteredSum[{j,city},j==x[i],...]  (a condition 

in a FilteredSum cannot depend on a decision variable). The only thing I could think of was: 

Minimize[Sum[{i,city},{j,city},{k,city},AsInt[j==x[i]]*AsInt[k==x[next[i]]]*dist[j,k ]]]  

Note that I used integers as set elements. I started with using data-binding through tables, using set elements 

{'city0','city1',...,'city13'}. This made the model somewhat more complicated, as x[i] is an integer. So the CSP 

formulation uses the simpler set {0,1,...,13} allowing us to operate on indices. (Some consider this bad modeling: in 

GAMS arithmetic on sets is actually discouraged, and needs a function ord() to convert the element τ always a 

string in GAMS τ to an integer). Unfortunately I was not able to solve the small 14 city example with the CSP 

solver using this formulation. (Even after adding City[0]==0  as constraint). To check the input, I also tried a MIP 

formulation with Gurobi. That solved this small instance just fine. 

In the MIP formulation we needed to formulate: 

FilteredSum[{i,city},i != 'city0', x['city0',i]]== 1 

From what I can see this is not possible: a set element not being an integer can not be used in OML. As a 

workaround I created sets by binding to some dummy parameters: 

Sum[{i 0,city0},{i,city2},x[i0,i]]==1  



 

where city0  is a set containing only a single element 'city0', and city2  is a set {'city1',...,'city13'}.  The 

constraint 

FilteredSum[{i,city},{j,city},i != j,x[i,j]] <= N  

did not work as i and j are not numeric. A workaround could be to have a parameter num[city]  which can be 

calculated in Excel and then imported. Then the condition can read: num[i]!=num[j] . I just used 

dist[i,j]>0  as condition, as that also can be used to exclude the diagonal (the Excel spreadsheet makes sure 

all diagonal distances dist[i,i]  are zero). A more general approach would be to be able to introduce sparse 2-

d sets (like one could do in AMPL and GAMS), but OML has only one dimensional sets as far as I know. 

This exercise was really meant to explore the expressiveness of OML. The little model actually shows some 

interesting issues with OML and some possible workarounds. I do not want to suggest this is a good way to solve 

TSP's. Obviously these approaches are not suited for large problems. A cutting plane algorithm often is quite 

effective for slightly larger problems (see this 42 city problem ). The Excel plug-in does not allow us to implement 

such an algorithm: an OML model can only contain a single model and has no looping facilities. For the real large problems, 

you need to look elsewhere, such as Concorde . 

The complete MIP model looks like: 

Model[  
 
// problem burma14 from tsplib  
 
  Parameters[Integers,N=14],  
  Parameters[Reals,f=0.1],  
  Parameters[Sets,city,city2,city0],  
  Parameters[Integers,dist[city,city]],  
  Parameters[Integers,dummy[city2]],  
  Parameters[Integers,dummy2[city0]],  
       
  
  Decisions[Integers[0,1],x[city,city]],  
  Decisions[Reals[0,Infinity],y[city,city]],  
 
  Constraints[  
      
     // Svestka formulation  
     Foreach[{i,city2}, FilteredSum[{j,city},dist[i,j]>0, y[j,i]] >= 1],  
     Foreach[{i,city2}, FilteredSum[{ j,city},dist[i,j]>0,y[i,j]] -  FilteredSum[{j,city},dist[i,j]>0, 
y[j,i]] == f],  
     FilteredSum[{i,city},{j,city}, dist[i,j]>0,x[i,j]] <= N,  
     FilteredForeach[{i,city},{j,city},dist[i,j]>0, y[i,j] <= (1 + N*f)*x[i,j]],  
 
     // tighten model  
     Foreac h[{i,city2},FilteredSum[{j,city},dist[i,j]>0, x[j,i]] == 1],  
     Foreach[{i,city2},FilteredSum[{j,city},dist[i,j]>0,x[i,j]] == FilteredSum[{j,city},dist[i,j]>0, 
x[j,i]]],  
     Sum[{i0,city0},{i,city2},x[i0,i]]==1,  
     Sum[{i0,city0},{i,city2},y[i0,i]]==1  
   ],  
 
 
  Goals[  
  Minimize[FilteredSum[{i,city},{j,city}, dist[i,j]>0,dist[i,j]*x[i,j]]]  
  ]  
 
  
]  

http://yetanothermathprogrammingconsultant.blogspot.com/2008/07/how-to-call-gams-from-access.html
http://www.tsp.gatech.edu/concorde.html


 

This formulation is from (Svestka, 1978). The problem is from TSPLIB and the distance calculations are replicated in 

the Excel spreadsheet (they involve complicated expressions due to projections). The final result is displayed in a 

chart: 

 

A slightly better picture results if we use decimal degrees. The coordinates above are in the format DD.MM 

(degrees and minutes). To convert this to decimal degrees we use the formula: DD+MM/60. The result is show 

below: 

 




















































































